

```

;-----
;- Скроллинг фона на 1 точку влево с помощью видеокарты МЕТЕОР-2013 www.zxkit.ru -
;- экран размером 320 x 240 точек, 256 цветов на точку -
;- спрайты фона размером 16 x 16 точек, карта уровня размером 15 x 256 спрайтов -
;-----
DRAW_SCR          ; НАСТРОЙКА ПАРАМЕТРОВ

    LD    IY, #5800          ; базовый адрес интерфейса видеокарты

    LD    A, (SCREEN_N)      ; каждое прерывание меняем местами
    XOR   3                  ; рабочий и отображаемый экраны
    LD    (SCREEN_N), A

    LD    (IY+1), SCR_NUM    ; установка номера экрана
    LD    (IY+2), A

    LD    (IY+1), SPRITES_ADDR0 ; адрес начала области спрайтов фона
    LD    (IY+2), 0
    LD    (IY+1), SPRITES_ADDR1
    LD    (IY+2), 0
    LD    (IY+1), SPRITES_ADDR2
    LD    (IY+2), 0

    LD    (IY+1), SIZE_X     ; размер спрайтов фона по-горизонтали
    LD    (IY+2), MAP_SXX    ; (16 или др.)

    LD    (IY+1), SIZE_Y     ; размер спрайтов фона по-вертикали
    LD    (IY+2), MAP_SSY    ; (16 или др.)

    LD    (IY+1), SPR_HN     ; старший байт номера спрайта
    LD    (IY+2), 0

    LD    (IY+1), SCR_HY     ; старший байт координаты Y
    LD    (IY+2), 0

    LD    E, PRINT_AUTOINC_X ; код команды компирования спрайта с автоинкрементом X

    LD    HL, (MAP_ADDR)     ; адрес начала карты уровня
    LD    A, (MAP_X)         ; текущая координата X на карте уровня
    LD    L, A

;-----
; ЦИКЛ ПО СТРОКАМ СПРАЙТОВ СВЕРХУ-ВНИЗ

    LD    B, 15              ; (240/16=15) кол-во спрайтов на экране по-вертикали
    LD    C, 0               ; координата Y строки спрайтов на экране = 0
DRW10
    LD    (IY+1), SCR_LY     ; установка мл. байта координаты Y спрайта на экране
    LD    (IY+2), C
    PUSH HL                  ; сохраняем координаты на карте 1 спрайта строки
;-----
; ПЕЧАТЬ СПРАЙТОВ В СТРОКЕ СЛЕВА-НАПРАВО

    LD    A, (MAP_DX)        ; текущий сдвиг спрайтов фона (0...15 или др.)
    NEG
    LD    D, #FF             ; добавляем знак «минус»
    JR    NZ, DRW20          ; если не 0 (сдвиг есть) - #FF в старший байт X
    INC   D                  ; иначе - 0 в старший байт X
DRW20
    LD    (IY+1), SCR_HX     ; установка ст. байта координаты X спрайта на экране
    LD    (IY+2), D
    LD    (IY+1), SCR_LX     ; установка мл. байта координаты X спрайта на экране
    LD    (IY+2), A

    LD    D, (HL)            ; загружаем номер спрайта из карты уровня
    LD    (#5801), DE        ; копируем 1-ый спрайт на экран
    INC   L                  ; адрес следующего спрайта на карте уровня

    LD    D, (HL)
    LD    (#5801), DE        ; копируем 2 спрайт на экран

```

```

INC     L

LD      D, (HL)
LD      (#5801), DE      ; копируем 3 спрайт на экран
INC     L

LD      D, (HL)
LD      (#5801), DE      ; копируем 4 спрайт на экран
INC     L

LD      D, (HL)
LD      (#5801), DE      ; копируем 5 спрайт на экран
INC     L

LD      D, (HL)
LD      (#5801), DE      ; копируем 6 спрайт на экран
INC     L

LD      D, (HL)
LD      (#5801), DE      ; копируем 7 спрайт на экран
INC     L

LD      D, (HL)
LD      (#5801), DE      ; копируем 8 спрайт на экран
INC     L

LD      D, (HL)
LD      (#5801), DE      ; копируем 9 спрайт на экран
INC     L

LD      D, (HL)
LD      (#5801), DE      ; копируем 10 спрайт на экран
INC     L

LD      D, (HL)
LD      (#5801), DE      ; копируем 11 спрайт на экран
INC     L

LD      D, (HL)
LD      (#5801), DE      ; копируем 12 спрайт на экран
INC     L

LD      D, (HL)
LD      (#5801), DE      ; копируем 13 спрайт на экран
INC     L

LD      D, (HL)
LD      (#5801), DE      ; копируем 14 спрайт на экран
INC     L

LD      D, (HL)
LD      (#5801), DE      ; копируем 15 спрайт на экран
INC     L

LD      D, (HL)
LD      (#5801), DE      ; копируем 16 спрайт на экран
INC     L

LD      D, (HL)
LD      (#5801), DE      ; копируем 17 спрайт на экран
INC     L

LD      D, (HL)
LD      (#5801), DE      ; копируем 18 спрайт на экран
INC     L

LD      D, (HL)
LD      (#5801), DE      ; копируем 19 спрайт на экран
INC     L

```

```

LD      D, (HL)
LD      (#5801), DE      ; копируем 20 спрайт на экран
INC     L

LD      D, (HL)
LD      (#5801), DE      ; копируем 21 спрайт на экран

; -----
;                ; ПОДГОТОВКА К ОТОБРАЖЕНИЮ СЛЕДУЮЩЕЙ СТРОКИ СПРАЙТОВ

POP     HL                ; восстанавливаем координаты на карте 1 спрайта строки
INC     H                ; переходим к следующей строке спрайтов на карте

LD      A, MAP_SSY        ; увеличиваем координату Y строки спрайтов на экране
ADD     A, C              ; (на 16 или др.)
LD      C, A

DJNZ    DRW10             ; следующая строка спрайтов до обнуления счетчика B

; -----
;                ; СДВИГ ЭКРАНА НА 1 ТОЧКУ — ИЗМЕНЕНИЕ КООРДИНАТ НА КАРТЕ

LD      A, (MAP_MAX_X)    ; достигли конца уровня ?
LD      L, A              ; (236 для режима 320x240, 240 для 256x192 или др.)
LD      A, (MAP_X)        ;
CP      L
JR      Z, DRW50          ; если да — дошли до конца, перестаем сдвигать фон

LD      A, (MAP_DX)       ; иначе — увеличиваем сдвиг спрайтов фона
INC     A
CP      MAP_S SX         ; сдвинули экран на целый спрайт ? (16 или др.)
JR      NZ, DRW40         ; если нет — координату X на уровне пока не меняем

DRW30   LD      A, (MAP_X) ; иначе — увеличиваем координату X на карте
INC     A
LD      (MAP_X), A        ; запоминаем новую координату X на карте уровня
XOR     A

DRW40   LD      (MAP_DX), A ; запоминаем новый сдвиг спрайтов фона

DRW50   RET               ; конец изображения фона игры

```